Laplace transforms in *Mathematica*

Write the ODE as an equation, and the initial conditions as a set of substitution rules:

```
myODE = y''[t] + 2 y'[t] + 15 y[t] == t Exp[-t]
```

$$15\,y(t)+2\,y'(t)+y''(t)=e^{-t}\,t$$

```
IC = {y[0] → 0, y'[0] → 1}
```

$$\{y(0)\to 0,\, y'(0)\to 1\}$$

Take Laplace transforms of both sides of the equation, and substitute the initial conditions into the equation.

```
ltODE = LaplaceTransform[myODE, t, s] /. IC
```

$$(\mathcal{L}_t[y(t)](s))\,s^2+2\,(\mathcal{L}_t[y(t)](s))\,s+15\,(\mathcal{L}_t[y(t)](s))-1=\frac{1}{(s+1)^2}$$

This equation will be easier to read if we write $Y(s)$ for $\mathcal{L}\{y(t)\}(s)$, which we can do using *a* substitution rule :

```
eqnForY = ltODE /. LaplaceTransform[y[t], t, s] → Y[s]
```

$$Y(s)\,s^2+2\,Y(s)\,s+15\,Y(s)-1=\frac{1}{(s+1)^2}$$

This is an algebraic equation for $Y(s)$ which we can solve

```
YSoln[s_] = Y[s] /. Solve[eqnForY, Y[s]][[1]]
```

$$\frac{s^2+2\,s+2}{(s+1)^2\,(s^2+2\,s+15)}$$

We've now computed the Laplace transform of the solution. Take its inverse Laplace transform to get the solution:

```
ySoln[t_] = InverseLaplaceTransform[YSoln[s], s, t]
```

$$\frac{1}{196}\,e^{-t}\left(14\,t+13\,\sqrt{14}\,\sin\left(\sqrt{14}\,t\right)\right)$$

Check the solution

```
ODECheck = myODE /. y → ySoln
```

$$-\frac{1}{98}\,e^{-t}\left(182\cos\left(\sqrt{14}\,t\right)+14\right)-\frac{13\,e^{-t}\sin\left(\sqrt{14}\,t\right)}{\sqrt{14}}+\frac{4}{49}\,e^{-t}\left(14\,t+13\,\sqrt{14}\,\sin\left(\sqrt{14}\,t\right)\right)+$$
$$2\left(\frac{1}{196}\,e^{-t}\left(182\cos\left(\sqrt{14}\,t\right)+14\right)-\frac{1}{196}\,e^{-t}\left(14\,t+13\,\sqrt{14}\,\sin\left(\sqrt{14}\,t\right)\right)\right)=e^{-t}\,t$$

```
FullSimplify[ODECheck]
```

True

```
ICCheck = {ySoln[0] == y[0], ySoln'[0] == y'[0]} /. IC
```

{True, True}